(delimited_expression ["(" ")" "[" "]" "\\{" "\\}" "|" "||"] @punctuation.bracket)
["{" "}"] @punctuation.bracket
["," ";"] @punctuation.delimiter
["^" "_" "'"] @punctuation.special

(number) @number
(symbol) @operator
(text) @string
(modifier) @method.call
(command) @function.macro

["sq" "rt" "/" "ncr"] @function.builtin
(environment
  . _ @keyword
  . _
  . _ @keyword
  .)

(
 (symbol) @variable
 (#any-of? @variable
  "a" "b" "c" "d" "e" "f" "g" "h" "i" "j" "k" "l" "m" "n"
  "o" "p" "q" "r" "s" "t" "u" "v" "w" "x" "y" "z"
  "A" "B" "C" "D" "E" "F" "G" "H" "I" "J" "K" "L" "M" "N"
  "O" "P" "Q" "R" "S" "T" "U" "V" "W" "X" "Y" "Z"
  "al" "be" "ga" "de" "ep" "ze" "et" "th" "io" "ka" "la" "mu"
  "nu" "xi" "pi" "rh" "si" "ta" "up" "ph" "ch" "ps" "om"
  "ep*" "ka*" "th*" "pi*" "rh*" "si*" "ph*" "ga*"
  "Al" "Be" "Ga" "De" "Ep" "Ze" "Et" "Th" "Io" "Ka" "La" "Mu"
  "Nu" "Xi" "Pi" "Rh" "Si" "Ta" "Up" "Ph" "Ch" "Ps" "Om"
  "Ga*" "De*" "Th*" "La*" "Xi*" "Pi*" "Si*" "Up*" "Ph*" "Ps*" "Om*"
  "dd" "pd" "na"
  "inf" "O/"
  ))

(
 (symbol) @type
 (#any-of? @type
  "CC" "HH" "NN" "PP" "QQ" "RR" "SS" "ZZ"
  ))

(
 (symbol) @function
 (#any-of? @function
  "div" "mod"
  "min" "max"
  "sin" "cos" "tan" "cot" "sec" "csc"
  "exp" "log" "ln"
  "det" "ker"
  ))
